# TRAIENT SUC D'UN TETRÀEDRE

(Estudi dels punts notables del tetràedre)


Autors del treball:

Joan Bertomeu Balagueró

Lluís Borràs Vila

Lluís Caballol Angelats

Maria Gas de Cid

Angels Miralles Verge

alumnes de C.O.U. de l'Institut de Batxillerat de Tortosa

Director del treball:

Antoni Gomà Nasarre, professor de Matemàtiques de l'Institut de Tortosa.


T O R T O S A , 1 9 8 3


INCENTRE
CIRCUMCENTRE
ORTOCENTRE
CENTRE
FIL
BARICENTRE

# INTRODUCCIÓ

Aquest treball ha estat executat per un equip de cinc alumnes de C.O.U. de l'I.B. "Joaquim Bau" de Tortosa. Hem comptat amb la valuosa col.laboració del nostre professor de Matemàtiques, senyor Antoni Gomà, que ens ha estimulat i dirigit en la investigació.

Les motivacions inicials que ens portaren a realitzar aquest estudi foren alguns dels interrogants que se'ns van plantejar al llarg de les classes de Matemàtiques durant l'estudi de la Geometria Analítica, com per exemple l'existència o no d'ortocentre en tots els tetràedres. Per aquesta raó, entre d'altres, el nostre treball està dedicat a la geometria del tetràedre, i més concretament, centrat en els punts notables d'aquest.

En relació a la matemàtica, és clara la utilitat d'alguns raonaments trets de la Geometria Sintètica, nogensmenys, la impossibilitat de caracteritzar-ho mitjançant una o l'altra via, ens ha portat a resoldre les qüestions plantejades per mig de la Informàtica. Així, l'elaboració d'un programa en llenguatge BASIC, que té com a finalitat l'esmentat estudi, ha esdevingut part central del treball, constituint a la vegada un esquema per a l'ordenació subsegüent.

Es ben cert que l'estudi té una finalitat eminentment - pràctica, però això no ens ha de portar a nosaltres ni a ningú a oblidar les qüestions teòriques, sense les quals tot estudi tècnic, científic i empíric resta sense la fermesa d'una base teòrica conseqüent, regla d'or que acompanya el mètode científic des de Galileu fins als nostres dies.

La recerca duta a terme en els folis següents a aquests ha estat acompanyada de les millors intencions i disposicions d'esperit científic.

# 0. Línia General del Programa

Hem començat el programa amb una sèrie de sentències d'inicialització que dimensionen les variables que utilitzarem al llarg del programa.

Cal fer l'aclariment de perquè no hem emprat les sentències "REM" per a encapçalar les diferents parts del programa. Hem prescindit d'aquestes sentèn - cies perquè al llarg del treball ja expliquem el contingut de cada part del progra ma en el moment oportú.

- ENTRADA DE DADES:

Per a introduir els vèrtexs del tetràedre a l'ordinador els hem nume- rat, i llavors hem utilitzat dos subíndexs, el primer d'ells ens indica el nombre del vèrtex, i el segon, la coordenada. El segon subíndex (1, 2, 3) correspon res- pectivament a les coordenades x, y, z. Per això dimensionem la variable O amb la sentència "DIM O(5,3)".




Lògicament pot semblar incor recte el haver posat el 5 com a primer - subíndex, però la raó que ens ha portat a incloure aquest cinquè vèrtex no ha estat la seva existència real com a un vèrtex - més, sinó que ho hem fet per a facilitar les operacions amb bucles i no haver de - fer més sentències de les estrictament ne cessàries, ja que l'únic que fem és donar els valors de les coordenades d'algun dels vèrtexs 1, 2, 3 ó 4 a aquest cinquè vèrtex. Aquest cinquè vèrtex només l'hem em - prat en dos casos, en el càlcul de l'ortocentre, i per fer el dibuix del tetràe - dre.

```
  1:TEXT :COLOR 1
 10:DIM O(5,3)
 20:DIM M(3,3)
 30:DIM G(3)
 40:DIM X(5):DIM Y
    (5):DIM Z(5)
100:FOR I=1TO 4
110:PAUSE "Vertex
    ";I;"  Coorden
    ades?"
120:FOR J=1TO 3
130:INPUT O(I,J)
140:NEXT J
150:LPRINT "Vertex
     ";I:LPRINT "
    (";O(I,1);",";
    O(I,2);",";O(I
    ,3);")"
160:NEXT I
170:LF 2
```

— SUBRUTINES:

Tots els punts notables del tetràedre que hem calculat els hem trobat com a intersecció de tres plans, excepte el baricentre, que l'hem fet per una fórmula. Per cadascun dels punts hem buscat els plans adequats, la intersecció dels quals ens dóna el punt buscat. Així, hem fet una subrutina que troba aquest punt. La utilitat de fer-ho d'aquesta manera és que un cop trobats els tres plans adequats hem utilitzat una altra subrutina per resoldre el sistema de tres equàcions i tres incognites que ens fa falta per a buscar el punt. La utilitat d'aquesta subrutina és clara, ja que la utilitzem tant per a l'ortocentre, per al circumcentre, per al punt que en diem filcentre i també per a l'incentre. Aquesta és la subrutina bàsica del programa.

Una altra subrutina que hem necessitat ha estat la que calcula determinants d'ordre tres i que ens permet resoldre el sistema d'equacions que formen les equacions que representen els tres plans. Aquests determinants d'ordre tres els haurem de calcular per a emprar la regla de Cramer per a resoldre el sistema.

Apart d'aquestes hi ha d'altres subrutines que es fan servir en els diversos apartats, i que ja explicarem allà:

- Per a dibuixar el tetràedre.
- Pera calcular l'equació dels plans de les cares.
- Per a comprovar quin és el pla bisector dels diedres.
- Per a calcular l'equació dels plans definidors del filcentre.
- Per a comprovar si el filcentre verifica totes les equacions.

SUBRUTINA DE RESOLUCIO D'UN SISTEMA DE TRES EQUACIONS AMB TRES INCOGNITES

El sistema és

$$\begin{cases} A(1,1)\,X(1)+A(1,2)\,X(2)+A(1,3)\,X(3)=B(1) \\ A(1,2)\,X(1)+A(2,2)\,X(2)+A(2,3)\,X(3)=B(2) \\ A(1,3)\,X(1)+A(2,3)\,X(2)+A(3,3)\,X(3)=B(3) \end{cases}$$

```
5000 FOR I=1 TO 3
5010 FOR J=1 TO 3
5020 LET M(I,J)=A(I,J)
5022 NEXT J
5024 NEXT I
5030 GOSUB 6000
5040 LET DETS=DET
5050 FOR J=1 TO 3
5060 FOR I=1 TO 3
5070 LET M(I,J)=B(I)
5080 IF J=1 THEN GOTO 5100
5090 LET M(I,J-1)=A(I,J-1)
5100 NEXT I
5110 GOSUB 6000
5120 LET D(J)=DET
5130 LET X(J)=D(J)/DETS
5140 NEXT J
5150 RETURN
```

SUBRUTINA DE CÀLCUL D'UN DETERMINANT 3X3

$$DET=\begin{vmatrix} M(1,1) & M(1,2) & M(1,3) \\ M(2,1) & M(2,2) & M(2,3) \\ M(3,1) & M(3,2) & M(3,3) \end{vmatrix}$$

```
6000 LET PO=M(1,1)*M(2,2)*M(3,3)
+M(2,1)*M(3,2)*M(1,3)+M(3,1)*M(1
,2)*M(2,3)
6010 LET NE=M(1,1)*M(3,2)*M(2,3)
+M(2,1)*M(1,2)*M(3,3)+M(3,1)*M(2
,2)*M(1,3)
6020 LET DET=PO-NE
6030 RETURN
```

# 1. És un Tetràedre ?

Abans de passar a fer tots els càlculs per a trobar els punts notables d'un tetràedre, cal assegurar-nos de que estem treballant realment sobre un tetràedre. El tetràedre l'introduïm a l'ordinador donant els vèrtexs, i per tant hem d'assegurar-nos de que els punts que introduïm no estan en un mateix pla.

Per a saber si és un tetràedre o no vam pensar en trobar el seu volum, si aquest es 0, lògicament no hi haurà tetràedre.

Hem vist a matemàtiques una fórmula bastant útil per a calcular el volum d'un tetràedre que consistia en fer un determinant d'ordre tres amb tres vectors concurrents un mateix vèrtex, i el resultat (en valor absolut) dividir-lo per 6.

Era el mètode més senzill que se'ns havia acudit per ha comprovar-ho, ja que així aprofitem una subrutina que ja teniem per a calcular determinants d'ordre tres, i al mateix temps calculavem ja el volum del tetràedre.

```
200:FOR J=1TO 3
210:FOR I=1TO 3
220:M(J,I)=O(1,I)-
    O(J+1,I)
230:NEXT I
240:NEXT J
250:GOSUB 6000
260:VOL=ABS (DET/6
    )
270:IF VOL=0THEN
    GOTO 9000
280:LPRINT "VOLUM
    TETRAEDRE : ";
    LPRINT VOL
290:LF 2
```

```
9000:LPRINT "ELS
     PUNTS DONATS
      NO FORMEN U
     N TETRAEDRE"
9010:END
```

```
10:GRAPH :GLCURSOR (70, -200):SORGN
20:LINE (0, 0)-(0, 140), 0, 1
30:LINE (0, 0)-(140, 0), 0, 1
40:LINE (0, 0)-(-70, -70), 0, 1
50:INPUT "Coordenades punt";A1, A2, A3
60:X=A1*6:Y=A2*15:Z=A3*15
70:LINE (-X, -X)-(Y-X, -X)-(Y, 0), 2, 0
80:LINE (Y-X, -X)-(Y-X, Z-X-2), 2, 0
90:LINE (Y-X-2, Z-X-2)-(Y-X+2, Z-X+2), 0, 1, B
100:CSIZE 1:GLCURSOR (Y-X+6, Z-X)
110:LPRINT "(";A1;", ";A2;", ";A3;")"
```







# 2. Dibuix del Tetràedre

El sistema que hem emprat a l'hora de dibuixar el tetràedre ha estat la perspectiva caballera, ja que hem cregut que era el que faria més senzill l'enteniment de la figura.

En aquest sistema, els eixos es situen de la següent forma: l'eix z en la vertical, l'eix y en l'horitzontal, i l'eix x formant 135º amb cadascun dels eixos anteriors.

A l'hora de fer el dibuix, i perquè aquest es semble més a la realitat, cal fer unes variacions en les dimensions, i així tenim uns coeficients de reducció que són de 1 als eixos y i z, i de 1/2 a l'eix x.

Per a dibuixar aquests eixos i els tetràedres, hem emprat la impresora del SHARP PC-1500, i per tant calia fer algunes adequacions en el sistema de coordenades. Abans que res cal esmentar que les coordenades d'aquest ordinador sobre les abcises tenen l'origen a l'extrem esquerre del paper, i augmenten cap a la dreta, tenint 216 posicions (unitats); sobre les ordenades, el tamany de les division és el mateix, i aquestes augmenten de dalt cap a baix.

Previ a la subrutina del programa que dibuixa els tetràedres, hem fet un petit programa que ens situa un punt qualsevol de l'espai tenint com a referència aquests eixos ( Al marge d'aquest full hem posat aquest programa acompanyat de dos exemples ).

Per a situar l'origen de coordenades dels eixos hem hagut de fer córrer el cursor 70 unitats cap a la dreta i 200 cap a baix, per a situar el nou origen en aquest punt. Una vegada fet això fem dibuixar els tres eixos de forma que les seves longituds siguin aproximada

```
310:FOR J=1TO 3
320:FOR I=1TO 4
330:G(J)=G(J)+O(I,
    J)
340:NEXT I
350:NEXT J
360:COLOR 3:LPRINT
    "BARICENTRE"
370:GOSUB 4000
380:X=6*G(1)/4:Y=1
    5*G(2)/4:Z=15*
    G(3)/4
390:LINE (Y-X-2,Z-
    X-2)-(Y-X+2,Z-
    X+2),0,3,B
395:GLCURSOR (-70,
    -100):TEXT
400:LPRINT "Coorde
    nades:":LPRINT
    "(";G(1)/4;",",
    ;G(2)/4;",";G(
    3)/4;")"
3999:END

4000:GRAPH :
     GLCURSOR (70
     ,-200):SORGN
4010:LINE (0,0)-(
     0,140),0,1
4020:LINE (0,0)-(
     140,0),0,1
4030:LINE (0,0)-(
     -70,-70),0,1
4040:FOR I=1TO 3
4050:O(5,I)=O(2,I
     )
4060:NEXT I
4080:FOR I=1TO 5
4090:X(I)=O(I,1)*
     6
4100:Y(I)=O(I,2)*
     15
4110:Z(I)=O(I,3)*
     15
4120:NEXT I
4130:FOR I=2TO 4
4140:LINE (Y(1)-X
     (1),Z(1)-X(1
     ))-(Y(I)-X(I
     ),Z(I)-X(I))
     ,0,0
4150:LINE (Y(I)-X
     (I),Z(I)-X(I
     ))-(Y(I+1)-X
     (I+1),Z(I+1)
     -X(I+1)),0,0
4160:NEXT I
4170:RETURN
```

ment les mateixes, i amb la inclinació deguda.

A l'hora de donar les coordenades del punt que volem dibuixar, al programa hem introduit una petita variació en les seves dimensions perquè les unitats no fossin excesivament petites. Així, fem equivaldre una unitat del nostre sistema damunt dels eixos y o z a 15 unitats del sistema de coordenades de l'ordinador, i una unitat sobre l'eix x a 8 unitats aproximadament dintre del sistema del PC-1500, mantenint més o menys d'aquesta forma, les proporcions de la perspectiva caballera.

Una vegada dibuixats els eixos, el que fem és trobar les coordenades del punt dins del sistema que utiliza l'ordinador tenint com a dades les coordenades a l'espai del punt. Així arribem a generalitzar les coordenades de qualsevol punt de l'espai, éssent aquestes dins del sistema de coordenades de l'ordinador : ( y - x, z - x )

- y : coordenada y del punt ( multiplicada per 15 per a adequar-la al sistema de l'ordinador.)
- z : coordenada z del punt ( multiplicada per 15 per a adequar-la al sistema de l'ordinador.)
- x : projecció de la coordenada x sobre qualsevol dels eixos ( la obtenim multiplicant la veritable coordenada per 6, així la reduïm segons la perspectiva, i l'augmentem per a adequar-la al sistema de l'ordinador )

Una vegada fet això amb un punt, dibuixar el tetràedre és fàcil, l'únic que cal fer és trobar els quatre punts que representen els quatre vèrtexs i unir-los entre si. Al marge hem ficat la subrutina que fa aquest dibuix i situa el baricentre del tetràedre.

```
Vertex  1
 ( 1,-1, 2)
Vertex  2
 (-2, 4, 9)
Vertex  3
 (-5, 7, 3)
Vertex  4
 ( 8, 6,-2)

VOLUM TETRAEDRE :
                 109

BARICENTRE
```



```
Coordenades
( 0.5, 4,
```

Cal dir que aquest cas és un dels que esmentem a l'apartat 0 en els quals emprem un cinquè vèrtex per a poder dibuixar el tetràedre amb un bucle, però aquest cinquè vèrtex només l'emprem per a una millor operativitat, ja que coincideix amb el segon vèrtex. Fent-ho així podem dibuixar dues ratlles seguides. En altres moments del programa hem preferit escriure més en lloc de fer un bucle, perquè resultava més sezill.

A continuació hem ficat quatre exemples de dibuixos d'un tetràedre amb el seu baricentre.

```
Vertex  1
 ( 0, 0, 0)
Vertex  2
 ( 4, 0, 0)
Vertex  3
 ( 0, 6, 0)
Vertex  4
 ( 2, 3, 6)

VOLUM TETRAEDRE :
                 24

BARICENTRE
```



```
Coordenades.
( 1.5, 2.25, 1.5)
```

```
Vertex  1
 ( 6, 1, 1)
Vertex  2
 ( 4, 0, 7)
Vertex  3
 (-1, 1,-1)
Vertex  4
 ( 5, 2, 3)

VOLUM TETRAEDRE :
      48.66666667

BARICENTRE
```



```
Coordenades
( 3.5,
```

```
Vertex  1
 ( 4, 3, 0)
Vertex  2
 ( 0, 0, 5)
Vertex  3
 ( 3,-1, 6)
Vertex  4
 ( 7, 6, 9)

VOLUM TETRAEDRE :

BARICENTRE
```



```
Coordenades.
( 3.5, 2, 5)
```

Cal fer una aclaració de perquè hem utilitzat un ordinador per a fer els dibuixos ( el SHARP PC-1500), i un altre per a la resta del programa ( el SINCLAIR ZX-81 ). La nostra intenció era fer-ho tot amb el SHARP PC-1500 , ja que aquest permet fer les representacions gràfiques necessàries, però la capacitat de la seva memòria és bastant limitada, i no hi ha cabut totes les variables necessàries. Per això ens hem vist obligats a fer el programa amb el SINCLAIR ZX - 81, amb més capacitat de memòria, però amb una impresora menys perfecta, i les subrutines per als dibuixos al SHARP PC - 1500. Per això hi ha algunes variacions en les instruccions que calculen els punts notables, ja que al no imprimir-los no calia fer les reduccions que ens permetrien fer el dibuix.

Per la reduïda capacitat de la memòria del SHARP PC-1500 només hem representat gràficament el baricentre. Amb la resta dels punts notables l'únic que caldria fer és repetir les mateixes operacions amb les seves coordenades, emprant la subrutina que ja tenim feta.

# 3. Baricentre

Entenem per baricentre d'un triangle el punt on es tallen les tres mitjanes del mateix. La mitjana del triangle és la recta que passa per un vèrtex, i el punt mig del costat oposat.

Generalitzant aquests conceptes per al tetràedre podem definir el baricentre d'un tetràedre com el punt on es tallen les seves quatre mitjanes. La mitjana d'un tetràedre és la recta que va d'un vèrtex al baricentre de la cara oposada. Aquest punt es pot trobar i per tant exiteix a qualsevol tetràedre.

CALCUL.

Escollint una referència adequada ( origen : un dels vèrtexs; vectors base : les tres arestes concurrents en aquest vèrtex ), podem arribar a generalitzar el càlcul de les coordenades del baricentre per a qualsevol tetràedre:




Sabem que el baricentre d'un triangle el podem trobar sumant les coordenades dels tres vèrtexs que el formen, i dividint-les per tres. Així trobem:

$G_A$(bar. de $\widehat{BCD}$) = (1/3, 1/3, 1/3)

$G_B$(bar. de $\widehat{ACD}$) = (0, 1/3, 1/3)

$G_C$(bar. de $\widehat{ABD}$) = (1/3, 1/3, 0)

$G_D$(bar. de $\widehat{ABC}$) = (1/3, 0, 1/3)

Trobem les medianes del tetràedre, i així arribem a trobar el següent sistema de vuit equacions, moltes de les quals estan repetides, que ha de ser compatible.

$$\begin{cases} x = y = z, \\ 3y = -x + 1, \\ x = z \\ y = z \\ x = y \\ 3y = -z - 1. \\ 3x = -y + 1 \end{cases}$$

Resolent aquest sistema veiem que es tallem totes en el punt G ( 1/4 , 1/4 , 1/4 ). Això vol dir que el vector que uneix l'origen de coordenades amb el baricentre dins de la referència triada és el vector ( 1/4 , 1/4 , 1/4 ).

Per tant, treballant en qualsevol referència i generalitzant-ho, podem trobar el baricentre de qualsevol tetràedre sumant les coordenades dels seus vèrtexs i dividint-ho per quatre.

COINCIDENCIA AMB EL CENTRE DE GRAVETAT VIST A FISICA

El nom de baricentre vol dir "centre de masses" o " centre de gravetat ". El que hem volgut fer ha estat el veure la concordància de la fórmula vista a Física amb el que hem fet a Matemàtiques.




$$\int_0^H \frac{m_i z_i}{M} \; ; \qquad A/\text{àrea} = \left(H/z\right)^2 = H^2/z^2$$

$$\int_0^H \frac{(\text{àrea})\,\rho\, dz \cdot z}{\frac{1}{3} A H \cdot \rho} = \frac{3z^2 \cdot \rho\, dz\, z}{H^2 A H \rho} \; ;$$

$$\int_0^H \frac{3z^3 \rho\, dz}{H^3 A \rho} = 3\left[\frac{z^4}{4H^3}\right]_0^H = \frac{3}{4}H$$

Fent que una altura del tetràedre coincideixi amb l'eix z, veiem que la coordenada z del baricentre surt a 3/4 de l'altura del tetràedre. Fent el mateix en les tres direccions, comprovariem la seva coincidència.

EXPLICACIO DEL PROGRAMA DEL BARICENTRE

Per a calcular el baricentre del tetràedre hem preferit trobar-lo segons la fórmula que hem deduït, és a dir sumant les coordenades i dividint-les per quatre.

Aquest programa ja l'hem vist quan parlavem del dibuix del tetraedre.

# 4. Circumcentre

Si ens preguntessim com generalitzariem el concepte de mediatriu del costat d'un triangle sens dubte buscariem el conjunt de punts que estan a la mateixa distància dels vèrtexs que defineixen els costats, això donaria un pla que se'n podria dir pla mediador.

També podriem pensar en el conjunt de punts que estan a la mateixa distància dels tres vèrtexs d'una cara, resultant llavors la solució una recta perpendicular à la cara pel seu circumcentre. La intersecció dels tres plans mediadors de les arestes d'una cara també ens donaria aquesta recta.

La intersecció de les quatre rectes perpendiculars a les quatre cares respectives pels seus respectius circumcentres seria un punt que estaria a la mateixa distància d'aquestes quatre cares, d'aquest punt en diem circumcentre, perquè és el centre d'una esfera que passa pels quatre vèrtex del tetràedre englobant-lo dintre d'ella; per tant ha de estar a la mateixa distància dels quatre vèrtexs del tetràedre.




De la següent manera es raonaria fàcilment la existència del circumcentre: Tendriem una recta R formada pel conjunt de punts equidistants d'uns vèrtexs 1,2,3, per exemple. Podriem considerar també el pla $\pi$, que consta del conjunt de punts que equidisten dels vértexs 3 i 4.

La recta R i el pla $\pi$ no poden esser mai paral.lels, per tant la seva intersecció ens donarà un punt que equidistarà de tots quatre vèrtexs

<u>EXPLICACIO DEL PROGRAMA DEL CIRCUMCENTRE</u>.

Una altra manera de trobar el circumcentre és fent la intersecció de tots els plans mediadors de totes les arestes ( total : 6 ), dels quals hem agafat tres que no pertanyiguen a la mateixa cara, la qual cosa és ben operativa de cara a la pràctica tal com es demostra en el programa subsegüen El haber agafat tres plans ens facilita el càlcul d'aquest punt notable mitjantçant la subrutina 5ØØØ, que serveix per calcular sistemes d'equacions ( en aquest cas plans mediadors ) que consten de tres equacions amb tres incognites.



```
430 FOR J=1 TO 3
440 FOR I=1 TO 3
460 LET Q(I)=(O(J,I)+O(J+1,I))/
2
470 LET C(I)=O(J,I)-O(J+1,I)
480 NEXT I
490 LET TI=C(1)*Q(1)+C(2)*Q(2)+
C(3)*Q(3)
500 FOR I=1 TO 3
510 LET A(J,I)=C(I)
520 NEXT I
530 LET B(J)=TI
540 NEXT J
550 GOSUB 5000
560 LPRINT "CIRCUMCENTRE=(";
X(1);",";X(2);",";X(3);")"
570 LPRINT
```

C(I) components dels vectors directors de les arestes

Q(I) coordenades punt mig de cada aresta

# 5. ORTOCENTRE

Entenem per ortocentre d'un triangle, el punt on es tallen les tres altures. Mirant de generalitzar aquest concepte per al tetràedre, definim com a tal el punt on es tallen les quatre altures. Aquestes quatre altures no sempre es tallaran en un punt. Definim com altura del tetràedre, la recta que passa per un vèrtex i és perpendicular a la cara oposada. Vam veure a classe alguns exemples de tetràedres no ortocèntrics, i més endavant, dins d'aquest estudi en veurem d'altres.

TEOREMES

— Les posicions relatives possibles de les altures d'un tetràedre són:

1.- Que no es tallin en cap punt, si cap de les arestes que els formen és ortogonal a la seva aresta oposada.

2.- Que es tallin per parelles si només hi ha un parell d'arestes ortogonals.

3.- Que es tallin totes quatre en un mateix punt, si cada aresta és ortogonal a la seva oposada. Es en aquest últim cas, únicament quan el tetràedre tindrà ortocentre.

— Que les parelles d'arestes oposades siguen ortogonals és condició necessaria i suficient per a que el tetràedre tingui ortocentre.

Demostració.— Si a un tetràedre les arestes oposades són ortogonals, podem fer un pla per a cada parell d'arestes de forma que sigui perpendicular a una d'elles i que tingui inclosa l'altra. En cas de que no siguin ortogonals, no es podrà fer aquest pla. Llavors la intersecció d'aquest pla amb el pla que forma la cara determinada per l'aresta a la qual és perpendicular, i l'extrem de l'altra aresta, determinarà una recta que serà una de les altures de la cara.




O = ORTOCENTRE

Com es veu al dibuix aquest pla que determinen A, B i M ha de ser perpendicular als plans BCD, i CD. Per tant, al pla ABM hi seran les altures del tetràedre que passen pels —

vèrtexs A i B respectivament, en estar en el mateix pla i no ser paral.leles s'han de tallar forçosament en un punt.

Fent això amb cada parell d'arestes, veuriem que les altures es tallen totes de dues en dues. Com que són quatre rectes que no estan en - un mateix pla, forçosament es tallaran totes en un punt: l'ortocentre.

— A tots els tetràedres que tenen ortocentre, les altures passen per l'ortocentre de la cara oposada. Si això es compleix - per a una altura, es complirà per a les altres. Es més, per a que un tetràedre tingui ortocentre s'ha de complir el que acabem d'esmentar.

<u>Demostració</u>.— Si sobre la cara del tetràedre que ara en direm base hi fem passar tres plans - perpendiculars que tinguin cadascun d'ells - inclosa una altura de la base, aquests plans es tallen formant una recta perpendicular a la base que passa pel seu ortocentre i qualsevol punt d'aquesta recta ens pot servir — com a vèrtex del tetràedre que compleixi el que diu el teorema.



— pla auxiliar.
— aresta del tetràedre
— altura de la cara.
— altura del tetràedre

Com es pot veure en el dibuix - cadascun dels plans auxiliars talla una de les altres cares del tetràedre de manera que forma una recta que és l'altura de la cara. Fent el mateix per a cada una de les cares quedarà a cada cara el dibuix del seu ortocentre provant-se així que el peu de cadasquna de les altures és l'ortocentre de la cara oposada.

El que més ens ha sobtat és el fet de que si es compleix per a una cara, es compleix per a totes. Això permet dibuixar tetràedres ortocèntrics amb suma facilitat, trobant l'ortocentre d'una cara i aixecant una - perpendicular que passe per ell.

## EXPLICACIO DEL PROGRAMA DE L'ORTOCENTRE.

En la part del programa dedicat a l'estudi de l'ortocentre hem calculat aquest com a intersecció de tres plans.

Es clar que previ al càlcul de l'ortocentre del tetràedre es té que comprovar la seva existència en un cas determinat.

La seva existència la vorem aplicant el teorema abans esmentat: " Que les parelles d'areste oposades siguin ortogonals és condició necessaria i suficient per a que el tetràedre tingui ortocentre".

condició de perpendicularitat esmentada en el paràgraf superior la comprovarem pent el producte escolar entre el vector d'una aresta (trobat restant les coordenades dels dos vèrtexs que uneix ), i de l'aresta oposada.

Si aquest producte ens dóna igual a 0 són perpendiculars.

Tenint ja la seguretat de l'existència de l'ortocentre en el tetràedre que estem estudiant, passem al seu càlcul. El realitzarem a partir de la intersecció de tres plans, que compleixen la condició de passar pel punt mig d'una aresta i ser perpendiculars a la mateixa. Amb la intersecció de dos d'aquests plans trobem una recta i amb la intersecció de tots tres trobem ja l'ortocentre del tetràedre.

Per a aquest càlcul s'ha de fer el sistema dels tres plans. Per això hem introduït una subrutina molt útil que ens calcula qualsevol sistema de tres equacions amb tres incognites, estalviant-nos fer el càlcul cada vegada que ens trobessem en un sistema de tres tipus.

```
700 FOR J=1 TO 3
710 LET F(1,J)=O(1,J)-O(3,J)
720 LET F(2,J)=O(4,J)-O(2,J)
730 LET F(3,J)=O(3,J)-O(2,J)
740 LET F(4,J)=O(1,J)-O(2,J)
750 LET F(5,J)=O(1,J)-O(4,J)
760 LET F(6,J)=O(3,J)-O(4,J)
770 NEXT J
780 FOR J=1 TO 3
790 LET V=V+F(4,J)*F(6,J)
800 LET W=W+F(1,J)*F(2,J)
810 NEXT J
820 IF V=0 AND W=0 THEN GOTO 90
0
830 LPRINT "NO TE ORTOCENTRE"
840 LPRINT
850 GOTO 1200
900 FOR M=1 TO 3
910 LET O(5,M)=O(1,M)
920 NEXT M
930 FOR J=1 TO 3
940 FOR I=1 TO 3
950 LET A(J,I)=O(J,I)-O(J+1,I)
960 NEXT I
970 LET B(J)=O(J+2,1)*A(J,1)+O(
J+2,2)*A(J,2)+O(J+2,3)*A(J,3)
980 NEXT J
990 GOSUB 5000
1000 LPRINT "ORTOCENTRE: (";X(
1);",";X(2);",";X(3);")"
1010 LPRINT
```

F(I,J) components dels vectors definidors de les arestes.

V, W productes escalars d'arestes oposades

RELACIO ENTRE L'ORTOCENTRE (si existeix), EL BARICENTRE I EL CIRCUMCENTRE

És conegut el fet que l'ortocentre, el circumcentre i el baricentre d'un triangle són alineats. El raonament vé de la consideració de dos triangles, construint un d'ells fent paral.leles als costats de l'altre pels seus vèrtexs.



Hem pensat si això també seria veritat en els tetràedres, naturalment en cas d'existir el punt ortocentre.

Hem vist que sí:

Si existeix ortocentre d'un tetràedre, en aquest tertàedre el baricentre, l'ortocentre i el circumcentre, són aliniats, amb el baricentre situat entre els altres dos.



L'esquema de la demostració és el següent:
Com que l'ortocentre, el circumcentre i el baqricentre de cada cara són aliniats, i en cas d'existir ortocentre del tetràedre l'altura del tetràedre passa per l'ortocentre de la cara oposada, resultarà que l'altura, la mitjana i la perpendicular al circucentre, determinaran un pla.
En aquest pla hauran de ser els tres punts notables de què ara parlem.
Com que això ho podrem repetir per cada cara, en la intersecció dáquests plans hauran de ser situats els punts, que, per tant, són aliniats.

# 6. Incentre

Podem definir-lo com aquell punt que compleix ésser el centre d'una esfera tangent a les cares del tetràedre, per tant la seva distància a les quatre cares és la mateixa.

Per a aconseguir el seu càl - cul ens hem fet servir de la intersecció= de tres plans bisectors .De fet,n'hi ha = sis,assegurant-nos la teoria, després, la= concordància del resultat (aproximat) amb les equacions restants.




Es convenient de fer un acla riment sobre aquests plans bisectors dels que en podem trobar dos per cada aresta ; un d'ells passa per la part exterior del= tetràedre i per tant no ens interessa: cal escollir el correcte. A l'hora de - triar-lo substituirem els vèrtexs de les arestes comprovant que el vèrtex superior és positiu mentres que el negatiu és l'inferior.




A → positiu.
B → negatiu.

A la vegada ens ha estat possible de limitar una recta bisectriu per cada vèrtex, de= la qual coneixem que els seus punts compleixen - que la distància a les tres cares és la mateixa, i que es calcula com la intersecció dels tres plans bisectors,passant per les arestes concu-- rrents en aquest vèrtex. Tenim una "bisectriu", és a dir, hem pogut generalitzar el concepte de bisectriu, entés normalment com la recta que està a distància igual de les dos rectes que formen un angle que= divideix en dues parts iguals.

L'incentre és un punt notable que podem trobar sempre en qualsevol tetràedre, la qual cosa és força evident ja que sempre tindrem possibilitat de trobar un punt interior al tetràedre que sigui a la mateixa distància de les - quatre cares. (*)

Ara bé, cal dir que no hi ha cap relació entre el punt on l'esfera és tangent a la cara i l'incentre d'aquesta, ja que són punts independents.

## EXPLICACIO DEL PROGRAMA DE L'INCENTRE

```
7000 LET E=C
7010 GOSUB 7500
7020 FOR U=1 TO 3
7030 LET R(U)=Q(U)/L
7040 NEXT U
7060 LET E=D
7070 GOSUB 7500
7080 FOR U=1 TO 3
7090 LET S(U)=Q(U)/L
7100 NEXT U
7120 FOR J=1 TO 3
7130 LET A(I,J)=R(J)-S(J)
7140 NEXT J
7150 LET B(I)=A(I,1)*O(A,1)+A(I,
2)*O(A,2)+A(I,3)*O(A,3)
7160 LET CC=A(I,1)*O(C,1)+A(I,2)
*O(C,2)+A(I,3)*O(C,3)-B(I)
7170 LET CD=A(I,1)*O(D,1)+A(I,2)
*O(D,2)+A(I,3)*O(D,3)-B(I)
7180 IF CC*CD<0 THEN GOTO 7230
7190 FOR J=1 TO 3
7200 LET A(I,J)=R(J)+S(J)
7210 NEXT J
7220 LET B(I)=A(I,1)*O(A,1)+A(I,
2)*O(A,2)+A(I,3)*O(A,3)
7230 RETURN
```

```
7500 FOR J=1 TO 2
7510 LET M(1,J)=O(A,J)
7520 LET M(2,J)=O(B,J)
7530 LET M(3,J)=O(E,J)
7540 NEXT J
7570 FOR U=1 TO 3
7580 LET M(U,3)=1
7590 NEXT U
7595 GOSUB 6000
7600 LET Q(3)=-DET
7610 LET M(1,2)=O(A,3)
7620 LET M(2,2)=O(B,3)
7630 LET M(3,2)=O(E,3)
7640 GOSUB 6000
7650 LET Q(2)=DET
7660 LET M(1,1)=O(A,3)
7670 LET M(2,1)=O(B,2)
7680 LET M(3,1)=O(E,2)
7690 GOSUB 6000
7700 LET Q(1)=-DET
7710 LET L=SQR (Q(1)*Q(1)+Q(2)*Q
(3)+Q(3)*Q(3))
7720 RETURN
```

En el programa, ens servim, per al càlcul de l'incentre, de= dues subrutines. En la primera (sentències 75ØØ - 771Ø ) calculem el - pla que passa pels punts genèrics A, B, E, i que al llarg del programa, - en les sentències 12ØØ - 136Ø, fem - variar de valor per a poder calcular els plans que passen pels punts O(1), O(2), O(3), O(4), escollits cada vegada de tres en tres.

La segona subrutina (sentències 7ØØØ - 722Ø ) la fem servir= per a calcular tres plans bisectors, i a la mateixa subrutina comprovem - que aquests plans siguin els que - ens cal considerar per al càlcul de- l'incentre pel mètode abans esmentat, de substituir a les equacions dels - plans bisectors els valors dels punts i comprovar el seu signe, i així deduir si treballem amb el pla correcte arranjant-ho si cal a les sentències 719Ø - 722Ø.

Una vegada tenim les equacions dels plans correctes, anem a la subrutina, situada a la sentència -

5ØØØ ,per a calcular sistemes d'equacions amb tres incògnites i així trobem l'incentre.

LLISTAT DE L'ESQUEMA DEL PROGRAMA DEL'INCENTRE

(que remet a les subrutines de la pàgina anterior)

```
1200 LET I=1
1210 LET A=1
1220 LET B=2
1230 LET C=3
1240 LET D=4
1250 GOSUB 7000
1260 LET I=2
1270 LET A=2
1280 LET B=3
1290 LET C=1
1300 LET D=4
1310 GOSUB 7000
1320 LET I=3
1330 LET A=3
1340 LET B=4
1350 LET C=1
1360 LET D=2
1370 GOSUB 7000
1380 GOSUB 5000
1390 LPRINT "INCENTRE:",,"(";X(1
);",";X(2);",";X(3);")"
```

Fem el pla bisector que passa per l'aresta o(A) o(B) i separa els vèrtexs o(C) i o(D)

(*) El raonament per a l'incentre seria el mateix que per al circumcentre canviant només la paraula vèrtex per cara.

# 7. Filcentre

Ja hem vist a l'apartat dedicat a l'incentre una generalització del concepte de bisectriu al tetràedre. Enteniem per "bisectriu" la recta que equidistava de les tres cares concurrents en un vèrtex.

Però hem trobat una altra possible recta que també podria entendre's com una altra generalització del concepte de bisectriu. Aquesta recta seria una recta equidistant de les tres arestes concurrents en un vèrtex. Per a cada parell d'arestes concurrents, podriem trobar el conjunt de punts que equidistessin d'aquestes, determinant així un pla perpendicular a la cara que determinen, passant per la bisectriu de l'angle que formen les dues-arestes. A aquest pla podriem anomenar-lo pla bisector entre dues arestes.

Així, a cada vèrtex trobariem com a interseeció dels tres plans bisectors a les arestes, una recta que equidistaria de les tres arestes, és a dir que estaria formada per tots els punts que equidistarien de les tres arestes.

Preguntant-nos si aquestes rectes que podriem definir per a cada vèrtex es trobarien en un punt, vam donar un nom a aquest punt hipotètic: FILCENTRE. Aquest està a la mateixa distància de les arestes que formen el tetràedre, i que és el centre d'una esfera que és tangent a les cares.

El nom de Filcentre ve donat pel fet de que a classe el senyor Gomà ens va fer imaginar l'existència d'aquest punt, fent que pensessim en un tetràedre format de filferro i l'esfera com una pilota que aniriem baixant fins que fos tangent a totes les arestes. Com el cetre de l'esfera, - era aquell punt pel que la pilota era tangent a les sis arestes de filferro que formaven el tetràedre, el nom que se'ns va acudir va ser el de filcentre.



No sempre existirà aquest punt, i vist això el que ens cal fer és cercar alguns teoremes que ens indiquin quan hi serà present aquest punt:

- Si el tetràedre és una piràmide regular la visió intuitiva ens permet assegurar que existeix filcentre

- El que es veia clar era que per a que el tetràedre tingués filcentre, els punts on l'esfera era tangent a tres arestes concurrents, les distàncies dels punts de tangència al vèrtex, havien de ser iguals. Així ens sortia el següent sistema d'equacions




$$\begin{cases} r = c + b \\ s = b + d \\ w = b + a \\ t = c + a \\ u = a + d \\ v = c + d \end{cases}$$

Operant podem arribar a trobar la següent condició:

Si hi ha filcentre $\Rightarrow r + u = s + t = v + w$

Aquesta és la condició que n'hem dit " Condició de la suma d'arestes oposades".

Així podem afirmar que perquè existeixi filcentre és necessari que les sumes de longituts d'arestes oposades coincideixin. Aquesta condició no podem assegurar que sigui suficient perquè el tetràedre tingue filcentre. Es a dir no hem pogut demostrar el teorema recíproc del primer que digués : -"Si les arestes oposades sumen igual, i poden ser les arestes d'un tetràedre ( caldria veure com saber si sis longituds poden ser les arestes d'un tetràedre ), llavors existeix filcentre ".

- Si existeix filcentre i tres arestes concurrents tenen iguals longituds , llavors el tetràedre és una piràmide regular. Això ho tenim demostrat amb raonaments trigonomètrics per a casos generals i aquest en particular. Però no els hem escrit perquè era més senzilla la seva demostració amb paraules basant-nos en la condició de la suma d'arestes oposades.




Suposarem que les tres arestes iguals concurrents són u, v i w.

Si u i v són iguals, i existeix filcentre, llavors s'ha de complir que r = s.

El mateix raonament el podem fer amb u i w, trobant així que:

r = s = t ,

i per tant el tetràedre és una piràmide regular.

– Si existeix filcentre i tres arestes concurrents formen el mateix angle, agafant dues qualsevol de les tres, llavors el tetràedre és una piràmide regular.

Per a demostrar aquest teorema ens hem vist obligats a emprar els raonaments que abans esmentàvem, ja que no hem trobat cap forma intuitiva de demostrar-ho.

Els càlculs són aquests:

Fent servir el dibuix de dalt i imaginant que els angles $\alpha$ i $\beta$ són iguals, arribarem a demostrar que u=v. Repetiríem el mateix raonament amb els altres angles veuríem que es tracta d'una piràmide regular.

Partint de u+r=v+s i aplicant als triangles d'arestes u,w,s i v,w,r el teorema del cosinus, arribem a $u+\sqrt{v^2+w^2-2vw\cos\alpha} = v+\sqrt{u^2+w^2-2uw\cos\beta}$

Si ara elevem al quadrat, després d'haver deixat sol a un costat un dels termes que tenen arrels quadrades, simplifiquem termes semblants, tornem a deixar sol un dels termes amb arrel i elevem al quadrat altra vegada, i tornem a simplificar, ens queda:

$$u^2w^2(1-\cos^2\alpha)+v^2w^2(1-\cos^2\beta)-2uvw^2(1-\cos\alpha\cos\beta)+(2u^2vw-2uv^2w)(\cos\alpha-\cos\beta)=0$$

D'aquesta igualtat segurament podríen sortir condicions generals per a l'existència de filcentre. De tota manera, si $\alpha=\beta$ , s'elimina el darrer sumand i queda $(1-\cos^2\alpha)(u^2+v^2-2uv)\,w^2=0$.

Això és $(1-\cos^2\alpha).(u-v)^2.w^2=0$.

Perquè això es compleixi només pot ser que u=v.

Com el que voliem era caracteritzar els tetràedres que tenien fil centre, vam estar buscant més teoremes possibles, i més concretament tractant de demostrar el recíproc del teorema al que haviem arribat abans, i que deia que la condició de la suma d'arestes oposades era necessaria per a tenir filcentre. Com no vam poder demostrar de cap forma, el que vam fer, va ser el.la borar un programa que ens permetria trobar molts d'exemples, i que potser ens farien arribar a alguna conclusió vàlida.

Vistos tots els exemples que hem fet, hem arribat a pensar en alguns possibles teoremes, i quan en trobavem algun, el que feiem era buscar al gun exemple que no ho complís. Així hem arribat a algunes conclusions:

- Només tenen filcentre les piràmides regulars. Aquest possible teorema l'hem rebutjat en trobar un tetràedre que no era piràmide regular, i que, a més, tenia filcentre. (Exemple 1)

```
VERTEX 1
(0,0,0)
VERTEX 2
(2,0,0)
VERTEX 3
(1,3.8729834,0)
VERTEX 4
(1,-0.77459667,3.7947332)

VOLUM TETRAEDRE: 4.8989795

BARICENTRE
(1,0.77459667,0.9486833)

CIRCUMCENTRE
(1,1.8073922,2.2135944)

NO TE ORTOCENTRE

INCENTRE:
(1,0.58083382,0.62221182)

ES COMPLEIX LA CONDICIO DE LA
SUMA D,ARESTES OPOSADES
*
*
*
*
FILCENTRE
(1,0.77459667,0.9486833)
```

Exemple 1




- Un altre possible teorema que se'n deduïa de les repeticions que es produïen en els exemples era: Si en un tetràedre existeix filcentre, llavors també existeix ortocentre. Aquest teorema també l'hem rebutjat en trobar un tetràedre amb filcentre que no tenia ortocentre. (Exemple 1)

- També ens hem preguntat si l'existència d'ortocentre implicava l'existència de filcentre, en veure una sèrie de coincidències en els exemples. Però hem trobat un tetràedre amb ortocentre, que no tenia filcentre. (Exemple 2)


VERTEX 1
(1,1,5)
VERTEX 2
(1,1,0)
VERTEX 3
(1,4,0)
VERTEX 4
(4,1,0)
VOLUM TETRAEDRE: 7.5
BARICENTRE
(1.75,1.75,1.25)
CIRCUMCENTRE
(2.5,2.5,2.5)
ORTOCENTRE
(1,1,0)
INCENTRE:
(1.7252983,1.7252983,0.7252983)
NO TE FILCENTRE

Exemple 2



A la vista de tots els exemples examinats creiem que es factible que sigui certa la proposició següent, que és la recíproca en certa manera de "la condició de suma d'arestes oposades".

- Si un tetràedre és tal que la suma de les seves arestes oposades coincideix per a les tres parelles d'arestes, el TETRAEDRE TE FILCENTRE.

Tanmateix, no hem pogut demostrar-ho.
Però els exemples següents, ben diversos, són una mostra que no és estrany que tinguem "esperances" que el teorema sigui cert.


VERTEX 1
(0,0,0)
VERTEX 2 (3,∅,∅)
(3,0,0)
VERTEX 3 (∅,4,∅)
[illegible]
VERTEX 4
(0,-0.5,3.966627)
VOLUM TETRAEDRE: 7.9372539
BARICENTRE
(0.75,0.875,0.99156[illegible])
CIRCUMCENTRE
(1.5,2,2.2677668)
NO TE ORTOCENTRE
INCENTRE:
(0.7455795,0.5575393[illegible],0.7455795)
ES COMPLEIX LA CONDICIO DE LA
SUMA D'ARESTES OPOSADES
FILCENTRE
(1,1,1.133893[illegible])

Exemple 3




Piràmide no regular.
Té, però, dues arestes concurrents (per parelles) de la mateixa longitud.
Alguna cara és triangle rectangle.

```
VERTEX 1
(0,0,0)
VERTEX 2
(0,4,0)
VERTEX 3
(4,3,0)
VERTEX 4
(0.54805899,3.1922359,4.9037892)
VOLUM TETRAEDRE:  13.076771
BARICENTRE
(1.1370148,2.548059,1.2259473)
CIRCUMCENTRE
(1.625,2,2.0379905)
NO TE ORTOCENTRE
INCENTRE:
(1.0561055,2.6245064,0.94464804)
ES COMPLEIX LA CONDICIO DE LA
SUMA D,ARESTES OPOSADES

FILCENTRE
(1.2192236,2.4364472,1.1962039)
```

Exemple 4




Exemple de piràmide no regular, amb dues arestes no concurrents de la mateixa longitud.

```
VERTEX 1
(0,0,0)
VERTEX 2
(3,0,0)
VERTEX 3
(0,4,0)
VERTEX 4
(-0.6666667,-1.5,5.7710964)

VOLUM TETRAEDRE:  11.542193

BARICENTRE
(0.58333333,0.625,1.4427741)

CIRCUMCENTRE
(1.5,2,3.8121005)

NO TE ORTOCENTRE

INCENTRE:
(0.70315012,0.61018253,0.7890527
9)
ES COMPLEIX LA CONDICIO DE LA
SUMA D,ARESTES OPOSADES

FILCENTRE
(1,1,1.4150979)
```

Exemple 5




Exemple de piràmide ~~no~~ regular, amb una cara triangle rectangle.




Exemple 6




Aquest darrer exemple ens sembla un tetràedre prou "estrany", per pensar que els tetràedres que compleixen la "condició de suma d'arestes oposades" tenen tots filcentre.

No ens hagués estat possible treballar amb tants d'exemples si no fos pel programa en llenguatge BASIC que hem treballat per a trobar el possible filcentre i comprovar, després, si realment ho és.

Vegem-lo. Abans, però, fem observar que el treball precisament amb l'exemple que tenim en aquesta pàgina ens ha fet fixar una permissió d'error en la "condició de la suma d'arestes oposades" pel fet de treballar els ordinadors, com és sabut, amb aproximacions dels nombres.

Doncs, hem canviat la sentència

```
1610 IF L(4)+L(6)=L(1)+L(2) AND L(1)+L(2)=L(3)+L(5) THEN GOTO 1640
```

per la següent:

```
1610 IF ABS ( L(4)+L(6)-L(1)-L(2))<10^-8 AND
     ABS(L(1)+L(2)-L(3)-L(5)) <10^-8 THEN GOTO 1640
```

EXPLICACIO DEL PROGRAMA DEL FILCENTRE.

Hem utilitzat dues subrutines englobades en les sentències que comencen a la 8ØØØ i a la 81ØØ.

A una d'elles ( la que va de la 8ØØØ a la 8Ø4Ø ) calculem els diferents coeficients i termes independents dels plans recercats, que fet i fet, per al seu càlcul directe els hem trobat definits pels vectors que formen el mateix angle amb les arestes, preses com a vectors, i que hem anat calculant canviant els diferents vèrtexs ( dos plans per cadascun . Es necessari que els dos vectors considerats en cada cas tinguin el mateix sentit, per això en algun cas ens hem vist obligats a canviar el sentit d'algun dels vectors-aresta.

En el programa apareixen algunes sentències ( LET I = 1, LET I = 2, LET I = 3 ) que pareix que quedarien millor dintre un bucle, però al igual que en alguna altra part del programa hem preferit de repetir sentències semblants per no fer-ho massa llarg, anat després a una subrutina en lloc d'un bucle, i són precisament les que defineixen el càlcul dels tres primers plans dels vuit totals que ens permeten calcular el filcentre, ja que els altres no més ratifiquen o exclueixen el possible filcentre.

Per això serveix l'altra subrutina : una vegada trobat el possible filcentre, la subrutina substitueix les seves coordenades en els altres plans. L'execució es detindrà en el moment en que la substitució del possible filcentre no doni 0, cosa que voldrà dir que no existeix filcentre.

Els asterics que apareixen quan es fa funcionar el programa responen a la justificació de cada pla que va ratificant el possible filcentre en questió. Cosa que no té més importància pel que fa a la part teòrica, però que és prou operatiu per a la visualització del programa.

SUBRUTINES

```
8000 FOR J=1 TO 3
8010 LET A(I,J)=F(B,J)-F(C,J)
8020 NEXT J
8030 LET B(I)=A(I,1)*O(U,1)+A(I,
2)*O(U,2)+A(I,3)*O(U,3)
8040 RETURN
8100 LET COM=A(3,1)*X(1)+A(3,2)*
X(2)+A(3,3)*X(3)-B(3)
8110 IF ABS (COM)>10**(-6) THEN
GOTO 1620
8120 RETURN
```

```
1580 FOR W=1 TO 6
1590 LET L(W)=SQR (F(W,1)*F(W,1)
+F(W,2)*F(W,2)+F(W,3)*F(W,3))
1600 NEXT W
1610 IF L(4)+L(6)=L(1)+L(2) AND
L(1)+L(2)=L(3)+L(5) THEN GOTO 16
40
1620 LPRINT "NO TE FILCENTRE"
1630 STOP
1640 LPRINT "ES COMPLEIX LA COND
ICIO DE LA    SUMA D,ARESTES OPOS
ADES"
1650 FOR W=1 TO 6
1660 FOR J=1 TO 3
1670 LET F(W,J)=F(W,J)/L(W)
1680 NEXT J
1690 NEXT W
1700 LET I=1
1710 LET V=1
1720 LET B=1
1730 LET C=4
1740 GOSUB 8000
1750 LET I=2
1760 LET C=5
1770 GOSUB 8000
1780 LET I=3
1790 LET V=2
1800 LET B=3
1810 LET C=4
1820 GOSUB 8000
1830 GOSUB 8100
1835 LET I=3
1840 LET B=2
1850 GOSUB 8000
1860 GOSUB 8100
1861 LPRINT "*"
1870 LET I=3
1880 LET V=3
1890 FOR J=1 TO 3
1900 LET F(1,J)=-F(1,J)
1910 NEXT J
1920 LET B=1
1930 LET C=3
1940 GOSUB 8000
1950 GOSUB 8100
1951 LPRINT "*"
1960 LET C=6
1970 GOSUB 8000
1980 GOSUB 8100
1981 LPRINT "*"
1990 LET V=4
2010 LET F(2,J)=-F(2,J)
2020 NEXT J
2030 LET B=5
2040 LET C=2
2050 GOSUB 8000
2060 GOSUB 8100
2061 LPRINT "*"
2070 LET C=6
2080 GOSUB 8000
2090 GOSUB 8100
2091 LPRINT "*"
2110 LPRINT "FILCENTRE",," (";X(1
);",";X(2);",";X(3);")"
```

Les arestes F(W) ja són definides en la part de l'ortocentre.

es fan de longitud 1 els "vectors aresta".

És fa el pla "bisector d'arestes" que passa pel vèrtex V i té d'arestes F(B) i F(C).

# Anex 1.

## Llistat general del programa

```
1 REM ENTRADA DE DADES
10 DIM O(5,3)
20 DIM M(3,3)
35 DIM F(6,3)
40 DIM A(3,3)
50 DIM B(3)
60 DIM C(3)
70 DIM D(3)
80 DIM Q(3)
90 DIM X(3)
91 DIM R(3)
92 DIM L(6)
93 DIM S(3)
94 LET W=0
95 LET V=0
100 FOR I=1 TO 4
110 PRINT "VERTEX ";I;" COORDEN
ADES?"
120 FOR J=1 TO 3
130 INPUT O(I,J)
140 NEXT J
150 LPRINT "VERTEX ";I;" (";O(I
,1);",";O(I,2);",";O(I,3);")"
160 NEXT I
170 LPRINT
```

```
199 REM ÉS UN TETRAEDRE
200 FOR J=1 TO 3
210 FOR I=1 TO 3
220 LET M(J,I)=O(1,I)-O(J+1,I)
230 NEXT I
240 NEXT J
250 GOSUB 5000
260 LET VOL=ABS (DET/6)
270 IF VOL=0 THEN GOTO 9000
280 LPRINT "VOLUM TETRAEDRE: "
;VOL
290 LPRINT
```

```
299 REM BARICENTRE
300 FOR J=1 TO 3
310 FOR I=1 TO 4
320 LET X(J)=X(J)+O(I,J)
330 NEXT I
340 LET X(J)=X(J)/4
350 NEXT J
360 LPRINT "BARICENTRE (";X(
1);",";X(2);",";X(3);")"
370 LPRINT
```

```
429 REM CIRCUMCENTRE
 430 FOR J=1 TO 3
 440 FOR I=1 TO 3
 460 LET Q(I)=(O(J,I)+O(J+1,I))/
2
 470 LET C(I)=O(J,I)-O(J+I,I)
 480 NEXT I
 490 LET TI=C(1)*Q(1)+C(2)*Q(2)+
C(3)*Q(3)
 500 FOR I=1 TO 3
 510 LET A(J,I)=C(I)
 520 NEXT I
 530 LET B(J)=TI
 540 NEXT J
 550 GOSUB 5000
 560 LPRINT "CIRCUMCENTRE",,"(";
X(1);",";X(2);",";X(3);")"
 570 LPRINT
```

```
699 REM ORTOCENTRE
 700 FOR J=1 TO 3
 710 LET F(1,J)=O(1,J)-O(3,J)
 720 LET F(2,J)=O(4,J)-O(2,J)
 730 LET F(3,J)=O(3,J)-O(2,J)
 740 LET F(4,J)=O(1,J)-O(2,J)
 750 LET F(5,J)=O(1,J)-O(4,J)
 760 LET F(6,J)=O(3,J)-O(4,J)
 770 NEXT J
 780 FOR J=1 TO 3
 790 LET V=V+F(4,J)*F(6,J)
 800 LET W=W+F(1,J)*F(2,J)
 810 NEXT J
 820 IF V=0 AND W=0 THEN GOTO 90
0
 830 LPRINT "NO TE ORTOCENTRE"
 840 LPRINT
 850 GOTO 1200
 900 FOR M=1 TO 3
 910 LET O(5,M)=O(1,M)
 920 NEXT M
 930 FOR J=1 TO 3
 940 FOR I=1 TO 3
 950 LET A(J,I)=O(J,I)-O(J+1,I)
 960 NEXT I
 970 LET B(J)=O(J+2,1)*A(J,1)+O(
J+2,2)*A(J,2)+O(J+2,3)*A(J,3)
 980 NEXT J
 990 GOSUB 5000
1000 LPRINT "ORTOCENTRE",,"(";X(
1);",";X(2);",";X(3);")"
1010 LPRINT
```

```
1199 REM INCENTRE
1200 LET I=1
1210 LET A=1
1220 LET B=2
1230 LET C=3
1240 LET D=4
1250 GOSUB 7000
1260 LET I=2
1270 LET A=2
1280 LET B=3
1290 LET C=1
1300 LET D=4
1310 GOSUB 7000
1320 LET I=3
1330 LET A=3
1340 LET B=4
1350 LET C=1
1360 LET D=2
1370 GOSUB 7000
1380 GOSUB 5000
1390 LPRINT "INCENTRE",,"(";X(1
);",";X(2);",";X(3);")"
```

```
1499 REM FILCENTRE
1580 FOR W=1 TO 6
1590 LET L(W)=SQR (F(W,1)*F(W,1)
+F(W,2)*F(W,2)+F(W,3)*F(W,3))
1600 NEXT W
1610 IF L(4)+L(6)=L(1)+L(2) AND
L(1)+L(2)=L(3)+L(5) THEN GOTO 16
40
1620 PRINT "NO HI HA FILCENTRE"
1630 STOP
1640 LPRINT "ES COMPLEIX LA CONDICIO DE LA    SUMA D,ARESTES OPOSADES"
1650 FOR W=1 TO 6
1660 FOR J=1 TO 3
1670 LET F(W,J)=F(W,J)/L(W)
1680 NEXT J
1690 NEXT W
1700 LET I=1
1710 LET V=1
1720 LET B=1
1730 LET C=4
1740 GOSUB 8000
1750 LET I=2
1760 LET C=5
1770 GOSUB 8000
1780 LET I=1
1790 LET V=2
1800 LET B=3
1810 LET C=4
1820 GOSUB 8000
1830 GOSUB 5000
1835 LET I=3
1840 LET B=2
1850 GOSUB 8000
1860 GOSUB 8100
1861 LPRINT "*"
1880 LET V=3
1890 FOR J=1 TO 3
1900 LET F(1,J)=-F(1,J)
1910 NEXT J
1920 LET B=1
1930 LET C=3
1940 GOSUB 8000
1950 GOSUB 8100
1951 LPRINT "*"
1960 LET C=6
1970 GOSUB 8000
1980 GOSUB 8100
1981 LPRINT "*"
1990 LET V=4
2000 FOR J=1 TO 3
2010 LET F(2,J)=-F(2,J)
2020 NEXT J
2030 LET B=5
2040 LET C=2
2050 GOSUB 8000
2060 GOSUB 8100
2061 LPRINT "*"
2070 LET C=6
2080 GOSUB 8000
2090 GOSUB 8100
2091 LPRINT "*"
2110 LPRINT "FILCENTRE",,"(";X(1
);",";X(2);",";X(3);")"
3999 STOP
```

```
4999 REM RESOLUCIÓ DE SISTEMES
5000 FOR I=1 TO 3
5010 FOR J=1 TO 3
5020 LET M(I,J)=A(I,J)
5022 NEXT J
5024 NEXT I
5030 GOSUB 6000
5040 LET DETS=DET
5050 FOR J=1 TO 3
5060 FOR I=1 TO 3
5070 LET M(I,J)=B(I)
5080 IF J=1 THEN GOTO 5100
5090 LET M(I,J-1)=A(I,J-1)
5100 NEXT I
5110 GOSUB 6000
5120 LET D(J)=DET
5130 LET X(J)=D(J)/DETS
5140 NEXT J
5150 RETURN
```

```
5999  REM  DETERMINANT  D'ORDRE  3
6000 LET PO=M(1,1)*M(2,2)*M(3,3)
+M(2,1)*M(3,2)*M(1,3)+M(3,1)*M(1
,2)*M(2,3)
6010 LET NE=M(1,1)*M(3,2)*M(2,3)
+M(2,1)*M(1,2)*M(3,3)+M(3,1)*M(2
,2)*M(1,3)
6020 LET DET=PO-NE
6030  RETURN


6999 REM  PLA  BISECTOR
7000 LET E=C
7010 GOSUB 7500
7020 FOR W=1 TO 3
7030 LET R(W)=O(W)/L
7040 NEXT W
7060 LET E=D
7070 GOSUB 7500
7080 FOR W=1 TO 3
7090 LET S(W)=O(W)/L
7100 NEXT W
7120 FOR J=1 TO 3
7130 LET A(I,J)=R(J)-S(J)
7140 NEXT J
7150 LET B(I)=A(I,1)*O(A,1)+A(I,
2)*O(A,2)+A(I,3)*O(A,3)
7160 LET CC=A(I,1)*O(C,1)+A(I,2)
*O(C,2)+A(I,3)*O(C,3)-B(I)
7170 LET CD=A(I,1)*O(D,1)+A(I,2)
*O(D,2)+A(I,3)*O(D,3)-B(I)
7180 IF CC*CD<0 THEN GOTO 7230
7190 FOR J=1 TO 3
7200 LET A(I,J)=R(J)+S(J)
7210 NEXT J
7220 LET B(I)=A(I,1)*O(A,1)+A(I,
2)*O(A,2)+A(I,3)*O(A,3)
7230 RETURN


7499 REM  EQUACIÓ  D'UNA  CARA
7500 FOR J=1 TO 2
7510 LET M(1,J)=O(A,J)
7520 LET M(2,J)=O(B,J)
7530 LET M(3,J)=O(E,J)
7540 NEXT J
7570 FOR W=1 TO 3
7580 LET M(W,3)=1
7590 NEXT W
7595 GOSUB 6000
7600 LET Q(3)=-DET
7610 LET M(1,2)=O(A,3)
7620 LET M(2,2)=O(B,3)
7630 LET M(3,2)=O(E,3)
7640 GOSUB 6000
7650 LET Q(2)=DET
7660 LET M(1,1)=O(A,2)
7670 LET M(2,1)=O(B,2)
7680 LET M(3,1)=O(E,2)
7690 GOSUB 6000
7700 LET Q(1)=-DET
7710 LET L=SQR (Q(1)*Q(1)+Q(2)*Q
(2)+Q(3)*Q(3))
7720 RETURN


7999 REM  SUBRUTINES  FILCENTRE
8000 FOR J=1 TO 3
8010 LET A(I,J)=F(B,J)-F(C,J)
8020 NEXT J
8030 LET B(I)=A(I,1)*O(U,1)+A(I,
2)*O(U,2)+A(I,3)*O(U,3)
8040 RETURN
8100 LET COM=A(3,1)*X(1)+A(3,2)*
X(2)+A(3,3)*X(3)-B(3)
8110 IF ABS (COM)>10**(-6) THEN
GOTO 1620
8120 RETURN

8999 REM  NO  ÉS  TETRÀERE
9000 LPRINT "ELS PUNTS DONATS NO
 FORMEN      TETRAEDRE"
9010 STOP
```

# Anex 2. Alguns exemples

```
VERTEX 1
(0,0,0)
VERTEX 2
(0,2,0)
VERTEX 3
(1.7320508,1,0)
VERTEX 4
(0.57735027,1,1.6329932)

VOLUM TETRAEDRE:   0.94280904

BARICENTRE
(0.57735027,1,0.40824829)

CIRCUMCENTRE
(0.57735027,1,0.40824829)

ORTOCENTRE
(0.57735027,1,0.40824829

INCENTRE:
(0.57735027,1,0.40824829)

ES COMPLEIX LA CONDICIO DE LA
SUMA D,ARESTES OPOSADES
*
*
*
*
*
FILCENTRE
(0.57735027,1,0.40824829)
```

Tetraedre regular



```
VERTEX 1
(5,5,5)
VERTEX 2
(-1,2,-4)
VERTEX 3
(2,4,6)
VERTEX 4
(4,-1,5)

VOLUM TETRAEDRE:   31

BARICENTRE
(2.5,2.5,3)

CIRCUMCENTRE
(2.5645161,2.3225806,0.51612903)

NO TE ORTOCENTRE

INCENTRE:
(3.0698149,3.1031989,4.4325984)
NO TE FILCENTRE
```



# 8. Temes Pendents

Al llarg de la realització del treball se'ns ha plantejat com un punt d'interés a tractar les condicions que han de complir uns certs nom - bres per a formar un tetràedre. Es evident que ja que cada aresta és comú a dos cares del tetràedre, ha de complir per a les dues la condició demanada= per a la formació d'un triangle: que la suma dels dos costats petits ens - done més gran que el costat més gran. Però donat que això s'escapava dels - objectius del treball, no hem entrat massa en el seu desenvolupament, i no= hem aprofundit ja en les relacions que existeixen entre aquests nombres i - les suposades condicions que haurien de complir. Això seria l'objectiu d'un altre treball dedicat principalment a l'estudi d'aquesta qüestió.

-o-o-o-o-o-o-o-

Un altre possible treball que es podria fer dins l'estudi dels= tetràedres és veure si existeix alguna relació entre els angles sòlids vis tos també a Física. Tema en el que no hem entrat perquè portaria un llarg es tudi que ens faria entrar en la Geometria Esfèrica, cosa que s'allunya cla- rament dels objectius del present treball.

-o-o-o-o-o-o-o-

La feina que hem desenvolupat al llarg del treball no h restrin git l'estudi al motiu principal d'aquest: punts notables del tetràedre, sinó que ens ha portat a descobrir noves relacions i suggeriments amb els que no hi haviem pensat com a conseqüència d'un estudi analític dels tetràedres. Així és com han sorgit les qüestions referents als angles que formen les ca res, els angles diedres, la construcció d'un tetràedre a partir d'unes da - des donades (coordenades).

Temes tots bons i prou interessants per desenvolupar un proper= estudi.

# 9. Material Emprat

- SHARP PC-1500 : Computador de butxaca amb impressora.

- SINCLAIR ZX-81 : Computador personal amb ampliació de memòria a 16 K i amb impressora.

# 10. Bibliografia


- Apunts de C.O.U. i d'Informàtica del Seminari de Matemàtiques de l'I.B. de Tortosa.
- Física de C.O.U. Editorial TEIDE
- Introducción a la Geometría. Eugenio ROYANES. Editorial ANAYA, Madrid, 1979.
- Lenguaje de programación BASIC. J. ASTOR VIGNAU. Ed. EUNIBAR
- ZX-81 BASIC Programming (Manual del ZX-81). Steven VICKERS. Sinclair Research Limited , London, 1980.
- SHARP PC-1500. Computador de bolsillo. Manual de manejo y programación.

# INDEX

Introducció .................................. I

0.-Línia general del programa ................ 1
1.-Es un tetràedre ........................... 3
2.-Dibuix del tetràedre ...................... 4
3.- Baricentre ............................... 8
4.- Circumcentre .............................10
5.- Ortocentre ...............................12
6.- Incentre .................................16
7.- Filcentre ................................19
Anex 1. Llistat general del programa ..........28
Anex 2. Alguns exemples .......................32
8.- Temes pendents ...........................33
9.-Material emprat ...........................34
10.- Bibliografia ............................34

Index .........................................35

# TRAIENT SUC D'UN TETRÀEDRE

ESTUDI DELS PUNTS NOTABLES DEL TETRÀEDRE

INSTITUT DE BATXILLERAT
"JOAQUIN BAU"
TORTOSA

Data: 29 abril 1983

Assumpte: Treball CIRIT

Per la present es fa constar que els alumnes

Joan Bertomeu Balagueró

Lluís Borràs i Vila

Lluís Caballol i Angelats

Maria Gas de Cid

Angels MIralles i Verge

estan matriculats durant el curs 1982-83 a l'Institut de Batxillerat de Tortosa, al Curs d'Orientació Universitària.

Que, dirigits pel professor Antoni Gomà i Nasarre han realitzat el treball "Traient suc d'un tetràedre - (estudi dels punts notables del tetràedre) - ", i que per a presentarlo al con curs de PREMIS PER A LA JOVENTUT, del CIRIT, es signa el present escrit a Tortosa, el vint-i-nou d'abril de 1983.

El Director,

Institut de Batxillerat "JOAQUIN BAU" - TORTOSA -

Signatures dels interessats,

Joan Bertomeu | Lluís Borràs | Lluís Caballol | Maria Gas | AngelsMiralles